Chipmunk BASIC v3.6.4(b8)

>load "/Applications/Chipmunk Basic stuff/Wavelength.bas"

10 rem Program to locatewavelength of equal reflective and emissive flux

12 rem Reflection of sunlight (6000 deg K) from Earth's surface

14 rem Thermal emission of Earth's surface at 300 deg K

16 rem Calculation of wavelength (x) using Newton Raphson Method

18 rem

20 rem Initial Conditions

25 c2 = 14386.8 : rem constant from Planck's Law

27 tsun = 6000 : rem deg K

30 tearth = 300 : rem deg K

35 rsun = 675000 : rem radius of sun in km

40 rearth = 150000000 : rem distance of Earth from Sun in km

45 albedo = 0.2 : rem Albedo of Earth's surface

50 rho = albedo*(rsun/rearth)^2

55 x = 3 : rem first guess

60 rem x=wavelength in micrometers

65 y = exp(c2/(tsun*x))-rho*exp(c2/(tearth*x))+rho-1

70 rem Y=function which equals zero when solution is found.

75 print x;y

80 rem

82 while abs(y) > 1.000000E-05

84 sun = exp(c2/(tsun*x)) : rem working variable

86 earth = exp(c2/(tearth*x)) : rem working variable

88 y = sun-rho*earth+rho-1

90 yprime = -c2*tsun*sun/((tsun*x)^2)+rho*c2*tearth*earth/((tearth*x)^2)

92 rem Yprime= derivative of Y with respect to x

94 x = x-(y/yprime)

96 print x;y

98 wend

99 end

>run

3 -34.240851

3.181765 -34.240851

3.377646 -13.10521

3.577542 -4.903546

3.756053 -1.730989

3.872978 -0.526874

3.911926 -0.108722

3.915317 -8.095250E-03

3.915341 -5.473097E-05

3.915341 -2.546327E-09

Therefore the answer is the crossover is at wavelength = 3.915  (m

